Tutorial Series – Drawing Traversals

In this article, I am going to discuss some of the thoughts and development processes behind writing a new macro.  The example may be considered a “Quick and Dirty” macro because there is no interface.

Please keep in mind that not all of the information you need is included in this article.  There may be a few cases where I jumped to the code that I used without a thorough explanation of how I go there.  The information provided here should be a good guideline for 

The Setup:

In this example, I will be writing a macro that traverses thru a SolidWorks drawing to collect and remove all of the revision triangles in the SolidWorks drawing.  Please note, the original writing if this macro was before SolidWorks implemented the revision tables and the revision markers that are linked back to the revision table.

If you would like to follow along with what I did, you can download the sample models and macros here().  Please note that this macro was written in SolidWorks 2007.  It has been tested in SolidWorks 2006 after the SolidWorks Type Libraries have been changed.  In the Macro Editor, click on the “References” menu and select the SolidWorks 2006 Type Libraries, then resave the macro.

Phase 1 – Discovery: Record a Macro:

For the purpose of writing this macro, a revision triangle will consist of a one or two character note with a triangular border.  This should be easy enough to find because it is usually unique from anything else in the drawing.  In this case, uniqueness is important so we don’t accidentally grab and delete something important.

What we want (and need) to know here is how SolidWorks sees these revision triangles.  Let’s see what the Macro Recorder can tell us.

· Start with a simple (or blank) drawing.

· Start the Macro Recorder

· Create a few (2-3) revision triangles on the drawing.  When you start creating the note, set your properties before placing it on the drawing.  Make sure you put at lease one annotation on the drawing sheet, and at least one on a drawing view. I changed the border for each note so I can what identifies the type of border.

· Select each revision triangle.  (NOTE: Do not select by a window.  We want to pick each individual revision triangle.  Window selection will pick up other drawing debris as well.)

· Press the “Delete” key to delete the selected revision triangles.

· Stop the macro recorder and save the macro.

Here is what I got!

' ******************************************************************************

' DelRevTriangles.swb - macro recorded on 09/18/07 by lkikstra

' ******************************************************************************

Dim swApp As Object

Dim Part As Object

Dim SelMgr As Object

Dim boolstatus As Boolean

Dim longstatus As Long, longwarnings As Long

Dim Feature As Object

Sub main()

Set swApp = Application.SldWorks

Set Part = swApp.ActiveDoc

Set SelMgr = Part.SelectionManager

boolstatus = Part.ActivateView("Drawing View1")

Dim Note As Object

Dim Annotation As Object

Dim TextFormat As Object

Set Note = Part.InsertNote("A")

If Not Note Is Nothing Then

   Note.Angle = 0

   boolstatus = Note.SetBalloon(2, 1)

   Set Annotation = Note.GetAnnotation()

   If Not Annotation Is Nothing Then

      longstatus = Annotation.SetLeader2(False, 0, True, True, False, False)

      boolstatus = Annotation.SetPosition(0.1168080375601, 0.1452269113924, 0)

      boolstatus = Annotation.SetTextFormat(0, True, TextFormat)

   End If

End If

Part.ClearSelection2 True

Part.WindowRedraw

Set Note = Part.InsertNote("A")

If Not Note Is Nothing Then

   Note.Angle = 0

   boolstatus = Note.SetBalloon(3, 1)

   Set Annotation = Note.GetAnnotation()

   If Not Annotation Is Nothing Then

      longstatus = Annotation.SetLeader2(False, 0, True, True, False, False)

      boolstatus = Annotation.SetPosition(0.1523260506329, 0.1356070632911, 0)

      boolstatus = Annotation.SetTextFormat(0, True, TextFormat)

   End If

End If

Part.ClearSelection2 True

Part.WindowRedraw

Set Note = Part.InsertNote("A")

If Not Note Is Nothing Then

   Note.Angle = 0

   boolstatus = Note.SetBalloon(1, 1)

   Set Annotation = Note.GetAnnotation()

   If Not Annotation Is Nothing Then

      longstatus = Annotation.SetLeader2(False, 0, True, True, False, False)

      boolstatus = Annotation.SetPosition(0.1439086835443, 0.1055450379747, 0)

      boolstatus = Annotation.SetTextFormat(0, True, TextFormat)

   End If

End If

Part.ClearSelection2 True

Part.WindowRedraw

boolstatus = Part.Extension.SelectByID2("DetailItem157@Drawing View1", "NOTE", 0.1090367341772, 0.1338033417722, 0, False, 0, Nothing, 0)

Part.EditDelete

boolstatus = Part.Extension.SelectByID2("DetailItem158@Drawing View1", "NOTE", 0.1270739493671, 0.1085512405063, 0, False, 0, Nothing, 0)

Part.EditDelete

End Sub
I can see that SolidWorks created the note first, then set the parameters (angle, balloon, leader, position, etc…) for the note.  By looking at the statement “boolstatus = Note.SetBalloon(3, 1)”, I can see that the first number is different for each note I placed because I created notes with different balloons.  This number represents the balloon type (triangle, hexagon, etc…).

At the bottom, where I deleted the notes, I can see that I need the Selection Manager.  What is also helpful here is the “Part.EditDelete”.  This tells how to delete the revision triangles once we found them.

I can also see the specific information about each selection I did, but these show the names of the notes, (“DetailItem152@Sheet1” and “DetailItem152@Drawing View1”), which also identifies whether the notes are on the drawing sheet or a drawing view.  Because the selection shows the names of the notes, we will have to develop another way to find and collect all of the revision triangles, since they can and will be different for each note on each drawing.  

Phase 2 – Discovery: How To Identify Revision Triangles:

First thing we need to do is define what we are looking for.  We discovered a lot about how a note is created by using the Macro Recorder.  We just need to reverse our thinking on how to find an existing note.

The next thing we need to look at is how a drawing file is put together.  When we look at the feature tree of a drawing, we can see that a drawing contains drawing sheets and views on each drawing sheet.  This is a clue that defines how we can search the drawing to find all of those revision triangles.  When we [Right-Click] on the drawing sheet and go to “Properties”, we can see a few settings for the sheet.  We also see that there may be a “Sheet Format/Size” associated with the drawing. When we [Right-Click] on the drawing view and go to “Properties”, we see some settings for the drawing view.  I don’t see anything here that will help us, but it is still a good idea to take a look.

Phase 3 – Discovery: Existing Examples:
A quick search for “Traverse Drawing” in the SolidWorks and Add-Ins API Help file reveals a few of useful examples.  

· The “Traverse Drawing FeatureManager Design Tree” example shows us how to access the views of the SolidWorks drawing.  Note that this example skips what the API finds is the first view of the drawing since it is the drawing sheet.  This example is not complete necause it does not traverse all views of the drawing.  It actually stops once it finds the data it’s looking for, in the first drawing view. 

· The “Select All Center Marks” example shows us how to traverse the views of the SolidWorks drawing.  

· The “Get Note By Name” example shows us how to get the note’s text. 

Keep in mind that we also need the ability to check the drawing sheets for revision triangles, and be able to traverse multi-sheet drawings.

Phase 4 – New macro for “universal” use:

The recorded macro shows us how a note was created, not how to find it.  The traversal examples do give us an idea on where to go, but they do not cover everything we need for creating a macro. Because of this, we need to start from scratch on this macro.

We know that a drawing has to exist before we can create (or find) a note, so that’s where we need to start.  The code below attaches to SolidWorks, then checks to see if there is an active document, then checks the type of document.  The MsgBoxes let the user know what was or was not found.

Sub Main()

  Set swApp = CreateObject("SldWorks.Application")         ' Attach to SWX

  Set Document = swApp.ActiveDoc                           ' Grab active doc

  If Document Is Nothing Then                              ' Is doc loaded

    MsgBox "No document loaded."                           ' Nothing - Warn

  Else                                                     ' Doc loaded"

    FileTyp = Document.GetType                             ' Get doc type

    If FileTyp = swDocDRAWING Then                         ' Is doc drawing ?

      ‘

      MsgBox "Drawing found.  Need drawing traversal routine."

      ‘

    End If                                                 ' Check doc type

  End If                                                   ' Check doc loaded

End Sub

So far, so good.  As we already discovered, a drawings contain sheets, sheets contain views, and both views and sheets can contain notes.  Looking at the “API Object Model tree” in the SolidWorks Add-Ins and API Help file shows a reference to “DrawingDoc” for SolidWorks drawings.  If we look at the DrawingDoc Object page in the help file, we can see that there are Methods, Properties, Events, and Accessors available for this object.  Under Methods, we can find “DrawingDoc.GetSheetCount” and “DrawingDoc.GetSheetNames”.  With this, we can start programming our traversal routine.  A little research reveals the code we need to traverse the sheets in a drawing.  Since sheet count and sheet names are available, we can use the For-next loop.

      SheetNames = Document.GetSheetNames                  ' Get sheet names

      For i = 0 To Document.GetSheetCount - 1              ' For each sheet

        Document.ActivateSheet (SheetNames(i))             ' Activate sheet

        ‘

        ‘ Sheet activated.
        ‘

      Next i                                               ' Next sheet

On the DrawingDoc Object page, under Methods, we can also find “DrawingDoc.GetFirstView” Because the view is considered an object, we need to set it as an object in the API.  Because we do not have a counter this time, we need to use the While-Wend loop and “View.GetNextView” to traverse the views.

        Set view = Document.GetFirstView                   ' Get first view

        While Not view Is Nothing                          ' View is valid

          ‘

          ‘ We got the view.
          ‘

          Set view = view.GetNextView                      ' Get next view

        Wend                                               ' View is valid

Since View is an object, we can look at the View Object page in the help file.  Under Methods here, we can find “View.GetFirstAnnotation3”.  We can also see that the Annotation (a note is a type of annotation) is also an object.  Our code to traverse the Annotations will be similar to our View traversal.

          Set Annotation = view.GetFirstAnnotation3        ' Get first annot

          While Not Annotation Is Nothing                  ' Annot valid?

            ‘

            ‘ We got the annotation.

            ‘

            Set Annotation = Annotation.GetNext2           ' Next annotation

          Wend                                             ' Annotation valid

Once we have the Annotation we need to determine the type.  Remember, we are looking for a “Note”.
          While Not Annotation Is Nothing                  ' Annot valid?

            If Annotation.GetType = swNote Then            ' Note annotations

            ‘

            ‘ The annotation is a note.

            ‘

            End If                                         ' Note annotation

            Set Annotation = Annotation.GetNext2           ' Next annotation

          Wend                                             ' Annotation valid

We can finally get information on the note to determine if it is one of the Revision Triangles we are looking for.  To do so, we need to check the border and how many characters are in the note.  
              Set NOTE = Annotation.GetSpecificAnnotation  ' Get specific note

              If NOTE.GetBalloonStyle = swBS_Triangle Then ' Triangle = Rev

                NoteVal = NOTE.GetText                     ' Get next note

                If Len(NoteVal) <= 2 Then                  ' 1-2 characters

                  ‘

                  ‘ We finally found a valid note.

                  ‘

                End If                                     ' No. char check

              End If                                       ' Triangle border?

Remember, we need to “select” the note before we can delete it. 
                  Annotation.Select2 True, 0               ' Add selection set

We also need a way to delete the revision triangles we found.  We can do this in a couple of ways.  We can delete the note as soon as we find it, but the action may give us a problem with our traversal routines.  I prefer to put the notes into a selection set and delete them after searching each view.  This can be done with the following statement.

                  Document.DeleteSelection False           ' Delete selection

Because our traversal “steps” several “levels” to the drawing to get to the annotation, we need to “nest” the routines (or embed each routine inside of the previous routine) to get everything to work properly.  All of this leads is to the following drawing traversal code.

      SheetNames = Document.GetSheetNames                  ' Get sheet names

      For i = 0 To Document.GetSheetCount - 1              ' For each sheet

        Document.ActivateSheet (SheetNames(i))             ' Activate sheet

        Set view = Document.GetFirstView                   ' Get first view

        While Not view Is Nothing                          ' View is valid

          ‘ Annotations in drawing view

          Set Annotation = view.GetFirstAnnotation2        ' Get first annot

          While Not Annotation Is Nothing                  ' Annot valid?

            If Annotation.GetType = swNote Then            ' Note annotations

              Set NOTE = Annotation.GetSpecificAnnotation  ' Get specific note

              If NOTE.GetBalloonStyle = swBS_Triangle Then ' Triangle = Rev

                NoteVal = NOTE.GetText                     ' Get next note

                If Len(NoteVal) <= 2 Then                  ' 1-2 characters

                  Annotation.Select2 True, 0               ' Add selection set

                End If                                     ' No. char check

              End If                                       ' Triangle border?

            End If                                         ' Note annotation

            Set Annotation = Annotation.GetNext2           ' Next annotation

          Wend                                             ' Annotation valid

          Document.DeleteSelection False                   ' Delete selection

          Set view = view.GetNextView                      ' Get next view

        Wend                                               ' View is valid

        ‘ Annotations on drawing sheet

        Set Annotation = Sheet.GetFirstAnnotation2         ' Get first annot

        While Not Annotation Is Nothing                    ' Annot valid?

          If Annotation.GetType = swNote Then              ' Note annotations

            Set NOTE = Annotation.GetSpecificAnnotation    ' Get specific note

            If NOTE.GetBalloonStyle = swBS_Triangle Then   ' Triangle = Rev

              NoteVal = NOTE.GetText                       ' Get next note

              If Len(NoteVal) <= 2 Then                    ' 1-2 characters

                Annotation.Select2 True, 0                 ' Add selection set

              End If                                       ' No. char check

            End If                                         ' Triangle border?

          End If                                           ' Note annotation

          Set Annotation = Annotation.GetNext2             ' Next annotation

      Next i                                               ' Next sheet

This code may be a bit more complicated than we first expected, but the clues we found while using the Macro Recorder, and examining the Feature Tree in the SolidWorks drawing, we were able to effectively determine the proper route to effectively traverse the drawing to find what we needed.
As programmed, the macro does work as expected. We could be done if this is all we need the macro to do.  The macro I released on my website has a counter that keeps track of the number of revision triangles that were found and deleted.  It also displays that count to the user just before the macro ends.
Phase 5 – Optional (Future) Enhancements:

This macro helped us develop a nice drawing traversal routine.  This is a reusable routine that can be the baseline for other macros where we need to find something in the drawing.  I have used this basic routine, with some additions, to check all annotations in a drawing, including dimensions, weld symbols, etc… to ensure the note fonts matched the default settings in the document.  This is one way to ensure the document meets your company’s CAD Standards.

Conclusion:

This is another tutorial that stems from a discussion that brought on a few comments like “My recorded macro does not work when I run it again.” or “How and where do I start writing a macro?” and “Where can I find help on writing a macro for SolidWorks?”

When you write a macro, follow these simple steps:

· Record a macro to “discover” API calls you may need.

· Study the recorded macro to “discover” how and when things happen.

· Upgrade the recorded macro to make it more “universal” and capable of working with other models and features.

· Add additional code to expand on the macro’s functionality. (ex: multiple pre-select)

· Add your user interface, if needed.  Again, start small, test functions, and work your way up.

· Add code to accommodate improper selections.

· Let it loose and see how others use the macro.

· Be ready to add updates as problems are “discovered”.

When is any macro actually done?  It can be done as soon as you get it to work as expected, or it may not be done until it has been “idiot-proofed” meaning that you have thought of, and added code to accommodate all of the potential ways the macro could be misused (intentional or not).  This decision is up to the developer, the user(s), and the environment the macro will be used (or abused) in.  Keep in mind that changes to the SolidWorks API, due to SolidWorks upgrades, may force you to make additional changes to the macro.

You can find a few more macros built on this concept by going to the “Quick and Dirty” (http://www.lennyworks.com/solidworks/default.asp?ID=44) macro page on my website.

